

Lab 3

This lab is built on the previous lab (Lab 2). *The next lab will be built on this lab, so please make sure that you go through all the instruction carefully and do all the assignments.* In this lab, you will learn how to:

- sketch domains and level sets,
- find limits of a function,
- find partial derivatives and directional derivatives,
- visualize tangent planes,
- visualize the gradient vectors.

1 To turn in

Do Problems 1-32 in a single Mathematica Notebook file, or ipynb file if you use JupyterLab. Write *your name* and *lab number* at the beginning of your report. Clearly label each problem to separate them from each other. Make sure to comment on each problem. If your code doesn't run correctly, explain what you are trying to do. Submit on Canvas both the *pdf file* and the *source file* (nb or ipynb). I will re-run all commands when I grade. Here is the breakdown of points:

Problems	Points
1, 4-6, 9, 14, 16, 18, 25, 27, 31	1
3, 8, 10-13, 15, 17, 21-24, 26, 28-30, 32	2
2, 7, 9, 19, 20	3
Readability of your report	3
Total: 32	Total: 62

2 Sketch the domain and level sets of a function

1. The command **RegionPlot** is used to draw a 2D region determined by one or more inequalities. For example, consider the function $f(x, y) = \sqrt{x^2 + y + 1}\sqrt{x - y - 1}$. The domain of this function is the set of points (x, y) such that $x^2 + y + 1 \geq 0$ and $x - y - 1 \geq 0$. To sketch this region, try the following:

```
RegionPlot[x^2 + y + 1 >= 0 && x - y - 1 >= 0, {x, -5, 5}, {y, -5, 5}]
```

This region has boundaries whose equations are $x^2 + y + 1 = 0$ and $x - y - 1 = 0$. You can highlight the boundaries of the region as follows:

```
p1=RegionPlot[x^2 + y + 1 >= 0 && x - y - 1 >= 0, {x, -5, 5}, {y, -5, 5}];
p2=ContourPlot[{x^2 + y + 1 == 0, x - y - 1 == 0},{x, -5, 5}, {y, -5, 5},
               PlotLegends->Automatic];
Show[p1,p2]
```

2. Can you sketch the region bounded by the four curves $xy = 1$, $xy = 3$, $x = y^2$, and $x = y^2 - 1$? Make sure to use both **RegionPlot** and **ContourPlot** commands as in the previous example.

3. The command **RegionPlot3D** is used to plot a 3D region determined by one or more inequalities. For example, consider the function $g(x, y, z) = \sqrt{x + y + z} \sqrt{4 - x^2 - y^2 - z^2}$. The domain of this function is the set of points (x, y, z) such that $x + y + z \geq 0$ and $x^2 + y^2 + z^2 \leq 4$.

```
RegionPlot3D[x + y + z >= 0 && x^2 + y^2 + z^2 <= 4, {x, -2, 2},
             {y, -2, 2}, {z, -2, 2}, AxesLabel -> {x, y, z}]
```

You can notice that the resolution of this hemisphere is not so good. Try adding the option `PlotPoints->40` (or some other large number) to the command.

4. To view the above solid in the first octant (where $x, y, z \geq 0$), you can narrow the range of x, y, z to the interval $[0, 2]$ instead of $[-2, 2]$. Please execute this.
5. The command **ContourPlot** is used to draw the level sets of a function of two variables. The command **ContourPlot3D** is used to plot the level sets of a function of three variables. For example, consider the function $f(x, y)$ given above. Define the function:

```
f[x_, y_] := Sqrt[x^2 + y + 1]*Sqrt[x - y - 1]
g[x_, y_, z_] := Sqrt[x + y + z]*Sqrt[4 - x^2 - y^2 - z^2]
```

6. Draw one level set:

```
ContourPlot[ f[x, y] == 0.5, {x, -5, 5}, {y, -5, 5}]
ContourPlot3D[ g[x, y, z] == 1.5, {x, -5, 5}, {y, -5, 5}]
```

7. The `ContourPlot/ContourPlot3D` commands can be used to plot any equation, such as

```
ContourPlot[x==y^2, {x,0,4}, {y,-2,2}]
ContourPlot3D[x+y==z^2, {x,-1,2}, {y,-1,2}, {z,0,2}]
```

The graph of a function can be plotted with `ContourPlot/ContourPlot3D`. For example, the function $h(x) = x^2$ is equivalent to the equation $y = x^2$, the function $k(x, y) = x^2 + y^2$ is equivalent to $z = x^2 + y^2$. Can you graph these two functions using `ContourPlot/ContourPlot3D`?

8. Draw multiple level sets at once (the *contour map*): the lighter the color corresponds to higher values of the function.

```
ContourPlot[ f[x, y], {x, -5, 5}, {y, -5, 5}]
ContourPlot[ f[x, y], {x, -5, 5}, {y, -5, 5}, Contours->50]
ContourPlot[ f[x, y], {x, -5, 5}, {y, -5, 5}, Contours->{0.2, 1, 2, 3, 4}]
ContourPlot3D[ g[x, y, z], {x, -5, 5}, {y, -5, 5}]
```

Can you explain what the second command and the third command do?

9. Plot the contour map (with 40 contours) of the function $f(x, y) = x \cos y - y \sin x$ on the square $x \in [3, 12]$, $y \in [2, 11]$. From the contour map, can you tell how many peaks and valleys the graph of f has? Explain your answer.
10. The solid bounded by four surfaces $z = 4 - y$, $z = 4 - x^2$, $z = 0$, $y = 0$ can be visualized by plotting all these surfaces together.

```
ContourPlot3D[{z==4-y, z==4-x^2, z==0, y==0}, {x, -2, 2}, {y, 0, 4}, {z, 0, 4}]
```

To enhance the visual effect, add the following options (separated by commas) to the previous command:

```
Mesh -> None
Boxed->False
AxesOrigin->{0,0,0}
PlotRange->{{-3,3},{-1,5},{0,5}}
AxesLabel->{Style["x",Bold,16], Style["y",Bold,16], Style["z",Bold,16]}
```

11. The surface $x^2y + y^2z + z^2x = 1$ is the 1-level set of the function $f(x, y, z) = x^2y + y^2z + z^2x$. Draw this surface.

3 Find limits of a function

12. To evaluate $\lim_{x \rightarrow 0} \frac{\cos x - 1}{x^2}$, try the following:

```
Limit[(Cos[x]-1)/x^2, x->0]
```

Can you graph the function to verify the answer?

13. To evaluate $\lim_{x \rightarrow \infty} (1 + \frac{1}{x})^x$, try the following:

```
Limit[(1+1/x)^x, x->Infinity]
```

Use the command `N` to evaluate this limit using 8 significant digits (review Lab 1 if you forget). Graph the function to verify the answer.

14. Find the limit

$$\lim_{x \rightarrow 0} \frac{e^x - e^{-x} - 2x}{x - \sin x}$$

Note that e^x is typeset as `E^x` or `Exp[x]`.

15. To evaluate the limit $\lim_{(x,y) \rightarrow (1,1)} \frac{x^3 - y^3}{x - y}$, try the following:

```
Limit[(x^3-y^3)/(x-y), {x,y}->{1,1}]
```

Graph the function to verify this answer. Note that you only need to look at the graph near the point $(x, y) = (1, 1)$.

16. Find the limit $\lim_{(x,y) \rightarrow (0,0)} \frac{xy}{x^2 + y^2}$.

17. To find the limit along a chosen path, we use the *replacement operator* `"/."`

```
Limit[x*y/(x^2-y^2)/.{x->0, y->t}, t->0]
Limit[x*y/(x^2+y^2)/.{x->t, y->2t}, t->0]
Limit[x*y/(x^2+y^2)/.{x->t, y->t+t^2}, t->0]
```

(x, y) goes to $(0, 0)$ along the y -axis in the first limit, along the line $y = 2x$ in the second limit, and along the parabola $y = x + x^2$ in the third limit. The part `/.{x->0, y->t}` means that x is replaced by 0 and y is replaced by t . The part `t->0` means that t is approaching 0 (very close to 0 but not equal to 0).

18. In the second limit of the previous problem, the point (x, y) travels along the line $y = 2x$. The corresponding point on the graph of $f(x, y) = \frac{xy}{x^2+y^2}$ travels along a 3D curve parametrized by

$$x = t, \quad y = 2t, \quad z = f(t, 2t)$$

You can graph this curve together with the surface as follows:

```
f[x_,y_] := x*y/(x^2+y^2);
p1=Plot3D[f[x,y],{x,-1,1},{y,-1,1},PlotPoints->25];
p2=ParametricPlot3D[{t,2t,f[t,2t]}, {t,-1,1}, PlotStyle->{Red,Thick}];
Show[p1,p2]
```

19. Find the limit of the function $\frac{xy}{x^2+y^2}$ as (x, y) goes to 0 along the sine curve $x = \frac{1}{2} \sin(5y)$. Plot the corresponding curve on the graph of $f(x, y) = \frac{xy}{x^2+y^2}$ as in Problem 18.
20. Explain why the limit

$$\lim_{(x,y) \rightarrow (-1,1)} \frac{y^2 - 1}{x + y}$$

does not exist. Hint: pick two different paths passing through $(-1, 1)$ which give two different limits.

4 Find partial derivatives and directional derivatives

21. To find the first, second, and third order derivatives of the function $\cos(x)$, try the following:

```
D[Cos[x], x]
D[Cos[x], {x, 2}]
D[Cos[x], {x, 3}]
```

Alternatively, you can give the function a name, say `f[x_] := Cos[x]`, and then run `f'[x]`, `f''[x]`, `f'''[x]`. The second derivative of $f(x)$ at $x = -\frac{\pi}{4}$ is computed by `f''[-Pi/4]`.

22. Find the third derivative of the function $f(x) = e^{\cos(x^2)}$. Evaluate numerically $f'''(1)$.
23. Use the following commands to compute partial derivatives $\frac{\partial f}{\partial x}$, $\frac{\partial^2 f}{\partial x \partial y}$, $\frac{\partial^2 f}{\partial x^2}$, $\frac{\partial^3 f}{\partial x \partial z^2}$ of a multi-variable function $f(x, y, z)$.

```
f[x_, y_, z_] := x^2 + y^3z + x*y*z^3
D[f[x,y,z], x]
D[f[x,y,z], x, y]
D[f[x,y,z], {x, 2}]
D[f[x,y,z], {x, 1}, {z, 2}]
```

24. Let $f(x, y, z) = x^{y^z}$. Check if $\frac{\partial^2 f}{\partial x \partial y} = \frac{\partial^2 f}{\partial y \partial x}$. Evaluate numerically $\frac{\partial^3 f}{\partial x^2 \partial y^3 \partial z^4}(1, 2, 3)$.

The derivative of a function $f(x, y)$ at (x_0, y_0) in the direction of vector v (not necessarily a unit vector) is given by $D_u f(x_0, y_0) = \nabla f(x_0, y_0) \cdot u$, where $u = \frac{v}{|v|}$ is the unit vector in the direction of v and $\nabla f = \left(\frac{\partial f}{\partial x}, \frac{\partial f}{\partial y} \right)$ is the gradient vector of f .

25. The gradient vector can be found using the command `Grad`. For example,

```
f[x_,y_] := x^2 + x*Sin[y]^2
Grad[f[x,y],{x,y}]
Grad[f[x,y],{x,y}]/.{x->1,y->Pi}
```

The last command is to evaluate the gradient of f at the point $(1, \pi)$.

26. Compute the derivative of the function $f(x, y, z) = \frac{x+y}{x+z^2}$ at the point $(1, 1, 1)$ in the direction of vector $v = (1, 2, 2)$.

5 Visualize tangent planes

27. The elliptic paraboloid $z = x^2 + 4y^2$ contains the point $(1, -1, 5)$. The tangent plane to the surface at this point is $z = -5 + 2x - 8y$. We want to draw the surface and the plane on the same picture. Try the following:

```
p1 = Plot3D[x^2+4y^2, {x,-1,3}, {y,-3,1}]
p2 = Plot3D[-5+2x-8y, {x,-1,3}, {y,-3,1}, PlotStyle->Blue]
Show[p1, p2, BoxRatios->1]
```

If you use Jupyter Notebook, add the prefix `Interact@` to the last command (`Show`) to be able to rotate the picture.

28. Draw the graph of the function $f(x, y) = \cos(x^2 + y) \sin(y)$ together with the tangent plane at $(x, y) = (1, 4)$.
29. Draw the surface $x^2y + y^2z + z^2x = 1$ together with the tangent plane to the surface at the point $(1, -1, 1)$.

6 Plot gradient vectors

A *vector field* is a map of arrows: *at each point on the plane or space is placed a vector*. To be more precise, a vector field is a function of two or three variables taking vector values. For example, the gradient of the function $f(x, y) = x^2 + y^2$, which is $\nabla f(x, y) = (2x, 2y)$, is a vector field because it assigns to each position (x, y) on the plane a vector $(2x, 2y)$. To plot a vector field, use the command **VectorPlot**.

30. Try the following:

```
VectorPlot[{2x, 2y}, {x, -2, 2}, {y, -2, 2}]
```

By default, Mathematica scales all the vectors so that they are of the same length while preserving the direction of each vector. The color reflects the actual length. The lighter the color, the longer the vector. To unscale the vectors, add the option `VectorScaling->Automatic` to the previous command. To make all vectors have the same color (blue by default), add the option `VectorColorFunction->None`. To make all vectors red, add the option `VectorStyle->Red`.

31. Plot the gradient vectors of the function $f(x, y) = \sin(xy)$ where $-4 \leq x \leq 4$ and $-4 \leq y \leq 4$.
32. Plot the contour map of the function f . Then use command **Show** to show the gradient vector field and the contour map on the same plot. Do the gradient vectors look perpendicular to the level sets? Notice that the gradient vectors point to the direction where f increases the most.